
SAS R© software has sections in its global symbol table for options andAbstract Description:
macro variables. Sysparm is both an option and a macro variable. As an
option, it can be assigned during startup on the command line; in programs,
values can be assigned with the options statement; values are stored and
referenced as a macro variable.

The purpose of this paper is to provide a general-purpose program, parse-Purpose:
sysparm.sas, which provides a method of deconstructing a list of comma-
separated values (csv) into separate macro variables. This is a useful
method of passing a set of parameter values from one program to another.

Introduction

This section covers the basics of macro variable assignment and referenc-Overview
ing syntax and the options assignment statement.
These are the topics in this section.

• macro variable assignment

• echo syntax

• options syntax

A macro variable assignment statement has five elements.macro variable
assignment 1. the verb %let

2. the name of the macro variable
3. an equals sign (=)
4. the value, which is all text between the equals sign and
5. the ending semicolon

The %put statement can be used to echo the value to the log.

%let mvar = text;

%put mvar = &mvar;

log: mvar = text

Since v9.3 this echo syntax is available.echo syntax

%let mvar = text;

%put echo &=mvar;

log: echo MVAR=text;

Sysparm can be assigned a value with either the macro variable assignmentoptions syntax
statement or the options statement.

%let sysparm = some text ;

%put echo mvar : &=sysparm ;

options sysparm = 'more text';

%put echo options: &=sysparm ;

log: echo mvar : SYSPARM=some text

echo options: SYSPARM=more text

Setup for Batch Processing

This section shows the files used in batch processing.Overview
These are the topics in this section.

• sas.cmd

• autoexec.sas

• my-program: my-program.bat, my-program.sas

• sysparm on command line

The batch files for each program depend on this sas.cmd file in the samesas.cmd
folder.

rem name: sas.cmd

"C:\program-files\SASHome\SASFoundation\9.4\sas.exe" %*

%* means pass all command-line parameters to SASNotes:

The batch files for each program shown here depend onautoexec.sas
filerefs project, and site inc with a path to the folder which contains the
site-include program parse-sysparm (p-sysprm.sas).

* name: autoexec.sas;

filename project '.'; *here;

filename site_inc '<...>\SAS-site\includes\';

Two files are required to submit a program in batch.my-program
1. my-program.bat and 2. my-program.sas.

rem name: my-program.bat

sas my-program

* name: my-program.sas;

*...;

This pair of batch and SAS programs show the sysparm command-linesysparm on command
line option used in the batch file and two SAS statements in the program used

to echo the value in the log.

rem name: demo-sysparm-command-line.bat

sas demo-sysparm-command-line -sysparm 'cmd-line'

*name: demo-sysparm-command-line.sas;

%put sysparm:%sysfunc(getoption(sysparm));

%put &=sysparm;

log: 1 %put sysparm:%sysfunc(getoption(sysparm));

sysparm:cmd-line

2 %put &=sysparm;

SYSPARM=cmd-line

Demonstration of Concepts

This is the overview, which consists of a list of topics in this section.Overview

• macro variable contains assignments

• scanning macro variable with delimiters

The value of a macro variable is always and only text and the text maymacro variable
contains assignment contain special characters. This program shows that the text may contain

a phrase of name=value. This value may be used in the macro variable
assignment statement.

*name: demo-mvar-eq-var-eq-text;

%let mvar=.;

%put echo &=mvar;

%let var_eq_value = mvar = value;

%put echo &=var_eq_value;

%let &var_eq_value;

%put echo &=mvar;

log: echo MVAR=.

4 %let var_eq_value = mvar = value;

5 %put echo &=var_eq_value;

echo VAR_EQ_VALUE=mvar = value

6 %let &var_eq_value;

7 %put echo &=mvar;

echo MVAR=value

This programs shows the use of the macro %scan function to fetch twoscanning macro
variable with a
delimiter

macro assignment statements from a delimited list.

*name: demo-using-scan.sas;

%let sysparm = b=2+c=3;

%put &=sysparm;

%let var_eq_text=%scan(&sysparm,1,+);

%put &=var_eq_text;

%let &var_eq_text;

%put &=b;

%let %scan(&sysparm,2,+);

%put &=c;

log: VAR_EQ_TEXT=b=2

5 %let &var_eq_text;

6 %put &=b;

B=2

7 %let %scan(&sysparm,2,+);

8 %put &=c;

C=3

Program Parse-Sysparm

This is the list of topics in this section.Overview

• parse-sysparm.sas

• parse-sysparm-test.sas

• parse-sysparm-test.log

This programs brings together the concepts shown previously.parse-sysparm

1. macro variable contains a delimited list
2. using scan function to separate set of tokens
3. assemble macro variable assignment statement

%put trace: p-sysprm=parse-sysparm beginning;

%put echo parameter: &=sysparm;

/* name: <UNC>\SAS-site\includes\p-sysprm.sas

parse-sysparm

description: convert sysparm to set of macro variables

purpose : subroutine for pararameterized includes

note : multiple values are named parameters

and delimited by comma: a=1,b=2

usage:

autoexec: filename site_inc '...\SAS-site\includes\';

options sysparm = 'a=1,b=2,d=4';

%include site_inc(p-sysprm);

**** ******/

DATA _null_;

attrib stmnt length = $%length(*let_&sysparm!);

if length(sysparm()) then do;

** upper bound is n(equal signs);

do i = 1 to countc(sysparm(),'=');

** assemble macro variable assignment statement;

stmnt= catx(' ','%let ',scan(sysparm(),i,','),';');

putlog 'echo: ' stmnt=;

call execute(cat('%nrstr(',stmnt,')'));

end;

end;

stop;

run;

%put trace: p-sysprm=parse-sysparm ending;

This program is used to test parse-sysparm.sas.parse-sysparm test

* name: <UNC>\SAS-site\sas-include-tests\parse-sysparm-test.sas;

options source2;

options sysparm = 'a1=1,b2=22,text=unquoted';

%include site_inc(p-sysprm);

options nonotes nosource nosource2;

options sysparm = 'data=sashelp.class,var=height,by=sex';

%include site_inc(p-sysprm);

options notes source;

%put _global_;

This is the log of the test program.parse-sysparm test
log

5 +%put trace: p-sysprm=parse-sysparm beginning;

trace: p-sysprm=parse-sysparm beginning

6 +%put echo parameter: &=sysparm;

echo parameter: SYSPARM=a1=1,b2=22,text=unquoted

[%included statements not shown]

echo: stmnt=%let a1=1 ;

echo: stmnt=%let b2=22 ;

echo: stmnt=%let text=unquoted ;

[some notes not shown]

NOTE: CALL EXECUTE generated line.

1 + %let a1=1 ;

2 + %let b2=22 ;

3 + %let text=unquoted ;

32 +%put trace: p-sysprm=parse-sysparm ending;

trace: p-sysprm=parse-sysparm ending

[some notes not shown]

trace: p-sysprm=parse-sysparm beginning

echo parameter: SYSPARM=data=sashelp.class,var=height,by=sex

echo: stmnt=%let data=sashelp.class ;

echo: stmnt=%let var=height ;

echo: stmnt=%let by=sex ;

trace: p-sysprm=parse-sysparm ending

65 %put _global_;

GLOBAL A1 1

GLOBAL B2 22

GLOBAL TEXT unquoted

GLOBAL DATA sashelp.class

GLOBAL VAR height

GLOBAL BY sex

Usage in Test-Driven Development (TDD)

This is the list of steps to development of programs in TDD.overview

1. make my-program.sas

(a) set defaults: %let ...

(b) include parse-sysparm: sysparm=blank, no replacement

(c) process

2. make my-program-test.sas with sysparm=(csv list)

3. make my-program-from-cmd-line.bat

TDD.1:
my-program.sas

*name: my-program.sas;

%let data = sashelp.bweight;

%let var = weight;

%include site_inc(p-sysprm);

PROC means data = &data;

var &var;

title3 &data &var;

run;

TDD.2:
my-program-test.sas

*name: my-program-test.sas;

*use defaults;

%include project(my-program);

*pass parameters to replace defaults;

%let sysparm=data=sashelp.class,var=weight;

%include project(my-program);

TDD.3:

rem name: my-program-from-cmd-line.bat

set data=data=sashelp.class

rem my-program-from-cmd-line.bat

call sas my-program -sysparm "%data%,var=height" -log class-height.log -print class-height.lst

call sas my-program -sysparm "%data%,var=weight" -log class-weight.log -print class-weight.lst

Many authors place several values in sysparm and use the scan function toSuggested Reading
separate them.

• Janka [3], and Yue [7], show colon-delimited values in sysparm;

• Coar [1], shows sysparm with values delimited by vbar parsed by
autoexec in batch programming;

• Langston [5], shows parsing of sysparm with comma-separated val-
ues;

• Salter and Cumming [6], shows how to assign default values when
sysparm is empty;

• Jackson [2], shows sysparm() in data step;

• Johnson [4], discusses issues in writing list-processing programs that
pass values from one step to later programs

Sysparm can be assigned by either a macro variable assignment statementConclusion
or an options statement. The option sysparm can be used on the command
line. In either case the value can contain one or more sets of assignment
tokens which can be parsed and copied into the global symbol table. This
method can be used to write a unit test of a program as well as run the
program with parameters set on the command line.

